An Analysis of Sugar Sources and their Effect on Blood Glucose Levels

Yale Quan
Valerie Torres
Bao Vu
California State University Long Beach

Table of Contents

Abstract

Sugar is derived from various sources and is widely used in food. Different types of sugar can have different effect on the body, whether positively or negatively. The focus of our experiment is to measure the percent change of blood glucose level after a subject consumes a specific source of sugar. The effect of different sources of sugar on blood glucose level is important to individuals who are sensitive to spikes (rapid increase) of blood glucose. These individuals include at risk populations such as those who have Type I or Type II diabetes, elderly individuals, or very young children.

In this experiment, the effects of 6 common sources of sugar on the blood sugar level of 4 subjects (2 Male, 2 Female) are measured. The sugar sources chosen are Raw Brown Sugar, Apple, Honey, Stevia Sweetener, White Rice, and Cheese. These were chosen because they are common in many diets. For instance, white rice is a stable in many diets and sugar is widely used for cooking and as sweetener for beverages such as coffee. For the purpose of this experiment, each subject will be given standard serving size of each of the sugar sources.

We will be measuring the percent change in in blood sugar of each subject experiences. The percent change will help control for a high initial blood glucose reading. The goal of our experiment is to see what types of sugar have a statistically significant effect on the mean blood glucose levels of the subjects.

Methodology

Blood glucose level will be measured using the *Oh'Care Lite Blood Glucose Monitoring System IGM-1003*. This meter will measure blood glucose levels using test strips. The test range is 20-600 mg/dL (milligrams per deciliter) with a sample volume of 0.5uL (microliters) of blood.

An initial reading will be taken before the subject is given the treatment. This initial reading will serve as a baseline reading and will be used to help calculate the percent change in blood glucose. The subject will be given the selected treatment and the sit still for 30 minutes. After 30 minutes has elapsed a second blood reading will the taken. We will call this second reading the post treatment reading. The data point recorded will be the percent change in blood glucose. This is calculated using the following formula:

$$Percent\ Change = \frac{Post\ Treatment\ Reading - Intial\ Reading}{Initial\ Reading} \times 100\%$$

Treatments will be given to subjects 2 hours apart. This time interval will allow the treatment sugar to be fully absorbed by the subject and allow a new initial reading that is not significantly influenced by any prior treatment.

The analysis for this experiment was performed in SAS software using PROC GLM.

Data Collected

Gender (Fixed Factor):

- Male
- Female

Subject (Nested Factor):

- Male Subject 1
- Male Subject 2
- Female Subject 1
- Female Subject 2

Sugar Source (Fixed Treatment):

1. **Source 1**: Natural Sucrose – 10g of cane sugar from 2 packets of Sugar in the Raw
2. **Source 2**: Natural Fructose – 99g of Gala Apple (one whole apple)
3. **Source 3**: Natural Glucose – 17g of Organic Honey
4. **Source 4**: Stevia Sweetener – 2g of Sugar Substitute from 2 Packets of Whole Earth Stevia
5. **Source 5**: Carbohydrates – 1 cup of cooked white rice
6. **Source 6**: Lactose – 60 grams of mozzarella cheese

	Male		**Female**	
Sugar Sources	**Subject 1**	**Subject 2**	**Subject 1**	**Subject 2**
Raw Brown Sugar (10g – 2 packets)	28.13	5.71	12.75	26.09
Apple (17g – 1 whole apple)	12.00	32.73	15.56	6.38
Honey (17g – 2 sticks)	5.68	10.20	14.12	0.00
Stevia (2g – 2 packets)	5.56	9.18	5.26	1.75
White Rice (1 cup cooked)	66.25	13.41	21.11	27.50
Mozzarella Cheese (60g – 2 sticks)	2.47	12.50	11.43	4.21

Model Statement and Assumptions

$$Y_{ijk} = \mu + \rho_{i(j)} + \alpha_j + \beta_k + (\alpha\beta)_{ij} + \epsilon_{ijk}$$

$i = 1, 2, 3, 4$ $\quad j = 1, 2$ $\quad k = 1, 2, 3, 4, 5, 6$

$\rho_{i(j)}$ = Random Subject Effect nested within Gender

α_j = Gender Fixed Effect

β_k = Sugar Source Fixed Effect

$(\alpha\beta)_{ij}$ = Interaction Effect between Sugar Source and Gender

$$\sum_{j=1}^{2} \alpha_j = 0;\ \sum_{k=1}^{6} \beta_k = 0;\ \sum_{j=1}^{2} (\alpha\beta)_{ij} = \sum_{k=1}^{6} (\alpha\beta)_{ij} = 0;\ \varepsilon_{ijk} \sim iidN(0, \sigma^2)$$

Model Checking / Residual Analysis

Prior to performing our analysis, we want to validate the assumptions of the analysis. The repeated measure analysis assumes that the residuals are normally distributed and have constant variance. To check the homogeneity of variance for each factor we performed Levene's test for Homogeneity of Variance on each factor of our experiment.

The hypothesis for Levene's Test for Homogeneity of Variance is $H_0: \sigma_1^2 = \sigma_2^2 = \cdots = \sigma_k^2$ and $H_a: \sigma_i^2 \neq \sigma_j^2$ for at least one pair (i, j). We perform the test at α = 0.05 and are interested in P-Values that are larger than 0.05. When the Levene's Test results in a P-Value larger than 0.05 we can conclude that the residuals for that factor have equal variance.

From Figure 1 we can see that the Levene's Test for Gender (Factor B) produces a F-Critical value of 1.25 with a corresponding P-Value of 0.2753 with 1 and 22 degrees of freedom. Since the P-Value is 0.2753 which is larger than α = 0.05 we conclude that the residuals for Gender have equal variance.

Levene's Test for Homogeneity of Y Variance ANOVA of Squared Deviations from Group Means					
Source	**DF**	**Sum of Squares**	**Mean Square**	**F Value**	**Pr > F**
B	1	289039	289039	1.25	0.2753
Error	22	5081055	230957		

Figure 1. Levene's Test for Gender

Figure 2 contains the SAS output for Levene's Test for Sugar Type (Factor C) produces a F-Critical value of 1.98 with a corresponding P-Value of 0.1310, which is larger than α = 0.05. Therefore, we conclude that the residuals for Sugar Type have equal variance.

Levene's Test for Homogeneity of Y Variance ANOVA of Squared Deviations from Group Means					
Source	**DF**	**Sum of Squares**	**Mean Square**	**F Value**	**Pr > F**
C	5	476060	95212.0	1.98	0.1310
Error	18	866459	48136.6		

Figure 2. Levene's Test for Sugar Type

Normality Tests and Plot

To check the normality of the residuals we perform the Shapiro-Wilk Test for Normality. The Shapiro-Wilk Test for Normality calculates the **W** statistic that test if a random sample, our residuals, come from a normal distribution. Small values of **W** are evidence that the random sample may not be normally distributed.

The hypothesis for this test is $H_0\text{:}\, The\ data\ comes\ from\ a\ normal\ distribution$ $H_a\text{:}\, The\ data\ may\ not\ come\ from\ a\ normal\ distribution$. We perform the test at α = 0.05

and are interested in P-Values that are larger than 0.05. When the Shapiro-Wilk Test results in a P-Value larger than 0.05 we can conclude that the residuals for the model come from a normal distribution.

From Figure 3 we can see that the Shapiro-Wilk test statistics **W** is 0.927 with a corresponding P-Value of 0.7344, which is larger than α = 0.05. Therefore, we can conclude that the residuals of our model are normally distributed.

Tests for Normality				
Test	**Statistic**		**p Value**	
Shapiro-Wilk	**W**	0.972732	**Pr < W**	0.7344

Figure 3. Test for Normality

We also generated a plot that plots the residuals against the Normal Quantile. We are looking for a linear pattern. A liner pattern indicates that the residuals are normally distributed. Figure 4 shows the result of this plot. We can see a relatively linear trend which indicates the residuals are normally distributed. From this plot and the results of the Shapiro-Wilk test we can conclude that the residuals are normally distributed.



Normal Score

Figure 4. Normality Plot of Residuals

Analysis

After validating the assumptions of our model, we proceed with our analysis. Figure 5, 6, and 7 displays the ANOVA table for the Model and the Factors using Type III SS. Our Model has a F-Statistic of 0.94 with a respective P-Value of 0.5533 with 13 and 10 degrees of freedom. Factor B (Gender) has a F-Critical value of 0.2780 with a P-Value of 0.2780 with 1 and 2 degrees of freedom. Factor A(B) (Subject nested in Gender) has a F-Critical value of 0.30 with a P-Value of 0.7438 with 2 and 10 degrees of freedom. Factor C (Sugar Type Treatment) has a F-Critical value of 1.94 with a corresponding P-Value of 0.1749 with 5 and 10 degrees of freedom. Lastly the interaction between B and C (Gender*Sugar) has a F-Critical value of 0.24 with a P-Value of 0.9336 with 5 and 10 degrees of freedom.

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	13	2539.199483	195.323037	0.94	0.5533
Error	10	2085.525167	208.552517		
Corrected Total	23	4624.724650			

Figure 5. ANOVA for Model and Error

Source	DF	Type III SS	Mean Square	F Value	Pr > F
A(B)	2	127.211633	63.605817	0.30	0.7438
C	5	2019.041350	403.808270	1.94	0.1749
B*C	5	254.418350	50.883670	0.24	0.9336

Figure 6. Hypothesis test for Factors using Type III SS

Tests of Hypotheses Using the Type III MS for A(B) as an Error Term					
Source	**DF**	**Type III SS**	**Mean Square**	**F Value**	**Pr > F**
B	1	138.5281500	138.5281500	2.18	0.2780

Figure 7. Hypothesis test for Factor B using A(B) as error

Interpretation of Analysis

From the results of the analysis we can conclude that Gender and Subject do not have a significant effect on the mean percent blood glucose level. Gender has a P-Value of 0.2780 and Subject has a P-Value of 0.7438. Both of those P-Values are larger than $\alpha = 0.05$ so we can conclude that these have no significant effect.

Our treatment (Factor C) has a P-Value of 0.1749 which is the closest to being significant and $\alpha = 0.05$. Since 0.1749 is larger than $\alpha = 0.05$ we conclude that our treatment had no significant effect on the mean percent blood glucose level.

Pair-Wise Comparison using Tukey

Tukey's method is used for simultaneous pairwise comparison between the level means of a factor. For a balanced sample size, the confidence level of the comparison is exactly $1-\alpha$. The Tukey method uses the *studentized range distribution* to calculate the minimum significant distance (MSD) between level means. If the absolute value of the difference between two level means exceeds the MSD, we conclude that the levels are significantly different.

Normally we would only perform pair-wise comparison using Tukey's method on factors that are significant. Since none of the factors in our experiment are significant, we will use Tukey's method to see if our treatment (Factor C) has at least one level that is pairwise significant. Figure 8 and 9 display the SAS output for pair-wise comparison using Tukey's method.

From Figure 8, we can see that Tukey's MSD is 35.467, a minimum of about 35 percentage point difference between two treatments' changes in blood glucose level. Figure 9 displays the difference between means between the levels of Factor C and the simultaneous confidence limits. We can see that none of the level means are significantly different. This reinforces the results of our ANOVA.

Alpha	0.05
Error Degrees of Freedom	10
Error Mean Square	208.5525
Critical Value of Studentized Range	4.91190
Minimum Significant Difference	35.467

Figure 8. Tukey's MSD Table

Comparisons significant at the 0.05 level are indicated by *.**				
C Comparison	**Difference Between Means**	**Simultaneous 95% Confidence Limits**		
5 - 6	24.42	-11.05	59.88	
1 - 5	-13.90	-49.36	21.57	
1 - 2	1.50	-33.96	36.97	
1 - 6	10.52	-24.95	45.98	
1 - 3	10.67	-24.80	46.14	
1 - 4	12.73	-22.73	48.20	
2 - 5	-15.40	-50.87	20.07	
2 - 6	9.02	-26.45	44.48	
2 - 3	9.17	-26.30	44.63	
2 - 4	11.23	-24.24	46.70	
3 - 5	-24.57	-60.03	10.90	
3 - 6	-0.15	-35.62	35.31	
3 - 4	2.06	-33.40	37.53	
4 - 5	-26.63	-62.10	8.84	
4 - 6	-2.22	-37.68	33.25	

Figure 9. Pairwise difference in means

Conclusion

We conducted this experiment to determine if different sugar sources (Raw Sugar, Apple, Honey, Stevia Sweetener, White Rice, and Cheese), given to subjects in serving size, have different effects on the subjects' blood sugar levels. While our experiment design and model met all the assumptions of the Analysis of Variance, our analysis of the data indicates that our treatment is not significant – different sugar sources do not have significant effect on the percentage change of the mean blood sugar level.

Although, we cannot conclude for certain that serving size of different sugar sources would yield approximately the same percent change in blood sugar level because our sample size is quite small. In order to further test our hypothesis, a similar experiment with larger sample size and more control is necessary. In addition, multiple glucose meters should be employed to account for the variability in blood sugar level readings from different devices.

```
data Sugars;
 input A B $ C Y;
 label
         A='Subject'
         B='Gender'
         C='Sugar'
         Y='Percent Change';
cards;
      1 M 1 28.13
      1 M 2 12.00
      1 M 3 5.68
      1 M 4 5.56
      1 M 5 66.25
      1 M 6 2.47
      2 M 1 5.71
      2 M 2 32.73
      2 M 3 10.20
      2 M 4 9.18
      2 M 5 13.41
      2 M 6 12.50
      1 F 1 12.75
      1 F 2 15.56
      1 F 3 14.12
      1 F 4 5.26
      1 F 5 21.11
      1 F 6 11.43
      2 F 1 26.09
      2 F 2 6.38
      2 F 3 0.00
      2 F 4 1.75
      2 F 5 27.50
      2 F 6 4.21
;

*************** Descriptive Statistics ****************;

proc means mean min max std data = Sugars;
      Title 'Descriptive Statistics for the Percent in Blood Sugar Change';
      var Y;
run;

proc sort data = Sugars out = boxplot_data;
      by A;
run;

proc boxplot data = boxplot_data;
      ods graphics off;
      Title 'Box Plot with Descriptive Statistics for Percent Blood Sugar
Change';
      plot Y*A;
      insetgroup min max mean/
            header = 'Extremes by Subject';
run;

proc sort data = Sugars out = boxplot_data;
      by B;
```

```
run;

proc boxplot data = boxplot_data;
      Title 'Box Plot with Descriptive Statistics for Percent Blood Sugar
Change';
      plot Y*B;
      insetgroup min max mean/
            header = 'Extremes by Gender';
run;

proc sort data = Sugars out = boxplot_data;
      by C;
run;

proc boxplot data = boxplot_data;
      Title 'Box Plot with Descriptive Statistics for Percent Blood Sugar
Change';
      plot Y*C;
      insetgroup min max mean/
            header = 'Extremes by Sugar Type';
run;

ods graphics on;

************** Levene Test using one way ANOVA and PROC GLM ***************;

proc glm data = Sugars;
      Title1 'Levene test for Gender';
      Title2 'A=Subject B=Gender C=Sugar Y=Percent Change';
      class B;
      model Y = B;
      means B/hovtest;
run;

proc glm data = Sugars;
      Title1'Levene test for Sugar Type';
      Title2 'A=Subject B=Gender C=Sugar Y=Percent Change';
      class C;
      model Y = C;
      means C/hovtest;
run;

******************* Repeated Measure Analysis *****************************;

proc glm data=Sugars;
      Title1 'Repeated Measure Analysis';
      Title2 'A=Subject B=Gender C=Sugar Y=Percent Change';
      class A B C;
      model Y = B A(B) C B*C;
      test h=B e=A(B);
      output out=sugar_out r=resid p=yhat;
run;

*************** Normality Tests and Plot ***************************;

proc univariate normal data = sugar_out;
```

```
    Title1 'Normal Tests for Sugar Experiment';
      Title2 'A=Subject B=Gender C=Sugar Y=Percent Change';
    var resid;
run;

proc rank normal=vw data = sugar_out;
      var resid;
      ranks nscore;
run;

proc plot;
      Title1 'Normal Plot for Sugar Experiment';
      Title2 'A=Subject B=Gender C=Sugar Y=Percent Change';
      plot resid*nscore='R';
      label nscore='Normal Score';
run;
quit;

****** Inference: Pairwise Multiple Comparison with Tukey *****************;

Proc GLM;
      Title1 'Tukey Multiple Comparison for Factor B';
      Title2 'A=Subject B=Gender C=Sugar Y=Percent Change';
      class A B C;
      model Y = B A(B) C B*C;
      means B / tukey cldiff;
run;

Proc GLM;
      Title1 'Tukey Multiple Comparison for Factor C';
      Title2 'A=Subject B=Gender C=Sugar Y=Percent Change';
      class A B C;
      model Y = B A(B) C B*C;
      means C / tukey cldiff;
run;

Proc GLM;
      Title1 'Estimated Mean and Standard Deviation of B(A)';
      Title2 'A=Subject B=Gender C=Sugar Y=Percent Change';
      class A B C;
      model Y = B A(B) C B*C;
      means A(B) / tukey cldiff;
run;
```

References

Montgomery, Douglas C. (2013). *Design and Analysis of Experiments 8th Edition*. Hoboken, NJ: Wiley Publishing

National Institute of Standards and Technology (2013, October 30) *Engineering Statistics Handbook*. Retrieved from: https://www.itl.nist.gov/div898/handbook/index.htm

SAS 9.3 User Guide (2011) *The GLM Procedure*. Retrieved from: https://support.sas.com/documentation/cdl/en/statug/63962/HTML/default/viewer.htm#glm_toc.htm